Lagrange Interpoláció, polinomfejlesztési algoritmus

A polinomot meghatárohatjuk a következô eljárással (minden egyes lineáris egyenletben pontosan egy ismeretlen (A)):

Az apriori alak: $p_{j+1} = p_j + A\ \omega$

X= {1, 2, 3, 4, 5}; Y= {0, 6, 24, 60, 120};

```
In[4]:= p0 = 0;

In[5]:= p1 = p0 + A (x - 1)
Out[5]= A (-1 + x)

In[6]:= p1 /. x → 2
Out[6]= A

In[7]:= Solve[(p1 /. x → 2) == 6]
Out[7]= {{A → 6}}

In[8]:= p1 = p0 + 6 (x - 1)
Out[8]= 6 (-1 + x)

In[9]:= p2 = p1 + A (x - 1) (x - 2)
Out[9]= 6 (-1 + x) + A (-2 + x) (-1 + x)

In[10]:= p2 /. x → 3
Out[10]= 12 + 2 A

In[11]:= Solve[(p2 /. x → 3) == 24]
Out[11]= {{A → 6}}

In[12]:= p2 = p1 + 6 (x - 1) (x - 2)
Out[12]= 6 (-1 + x) + 6 (-2 + x) (-1 + x)

In[13]:= p3 = p2 + A (x - 1) (x - 2) (x - 3)
Out[13]= 6 (-1 + x) + 6 (-2 + x) (-1 + x) + A (-3 + x) (-2 + x) (-1 + x)

In[14]:= Solve[(p3 /. x → 4) == 60]
Out[14]= {{A → 1}}

In[15]:= p3 = p2 + 1 (x - 1) (x - 2) (x - 3)
Out[15]= 6 (-1 + x) + 6 (-2 + x) (-1 + x) + (-3 + x) (-2 + x) (-1 + x)

In[16]:= p4 = p3 + A (x - 1) (x - 2) (x - 3) (x - 4)
Out[16]= 6 (-1 + x) + 6 (-2 + x) (-1 + x) + (-3 + x) (-2 + x) (-1 + x) + A (-4 + x) (-3 + x) (-2 + x) (-1 + x)
```

```
In[17]:= Solve[(p4 /. x → 5) == 120, A]

Out[17]= {{A → 0}}

In[18]:= p4 = p3 + 0 (x - 1) (x - 2) (x - 3) (x - 4)

Out[18]= 6 (-1 + x) + 6 (-2 + x) (-1 + x) + (-3 + x) (-2 + x) (-1 + x)

In[19]:= Expand[%]

Out[19]= -x + x^3

In[25]:= X = {1, 2, 3};
         F = {2, 5, 10};

In[27]:= Clear[LP];
         LP[0] := (P = F[[1]]);

In[29]:= LP[n_] := Module[{pl}, pl = P + A Product[x - X[[j]], {j, 1, n}];
           P = pl /. Solve[(pl /. x → X[[n + 1]]) == F[[n + 1]], A][[1]];
           {pl, (pl /. x → X[[n + 1]]) == F[[n + 1]],
            Solve[(pl /. x → X[[n + 1]]) == F[[n + 1]], A][[1]], P, Expand[P]}]

In[30]:= Manipulate[TableForm[{{LP[n]}}], {{n, 0}, 0, 2, 1}]

Out[30]=
 n
 2 + 3 (-1 + x) + A (-2 + x) (-1 + x)
 8 + 2 A == 10
 A → 1
 2 + 3 (-1 + x) + (-2 + x) (-1 + x)
 1 + x^2
```

Newton előállítása osztott differenciákkal (rekurzív)

A Lagrange alaknál új adat hozzávételére nincs lehetôség ⟶ Newton féle elôállítás osztott differenciák segítségével

```
In[31]:= Clear[OD, x0, y0, x1, y1]

In[32]:= NewtInterp[x_List, y_List, var_] := Sum[OD[Take[x, j], Take[y, j]] Product[(var - x[[i]]), {i, 1, j-1}], {j, 1, Length[x]}]
```

$$\text{NewtInterp}[x\_List,\ y\_List,\ var\_] := \sum_{j=1}^{\text{Length}[x]} \text{OD}[\text{Take}[x, j], \text{Take}[y, j]] \prod_{i=1}^{j-1} (var - x[[i]])$$

```
In[33]:= NewtInterp[{x0, x1, x2}, {y0, y1, y2}, x]

Out[33]= OD[{x0}, {y0}] + (x - x0) OD[{x0, x1}, {y0, y1}] + (x - x0) (x - x1) OD[{x0, x1, x2}, {y0, y1, y2}]

In[34]:=
         OD[{x_}, {y_}] := y;
         OD[x_List, y_List] :=
          (OD[Drop[x, 1], Drop[y, 1]] - OD[Drop[x, -1], Drop[y, -1]]) / (Last[x] - First[x])

In[36]:= OD[{x0, x1}, {y0, y1}]
```

Out[36]= $\dfrac{-y0 + y1}{-x0 + x1}$

In[37]:=
```
NewtInterp[{x0, x1, x2}, {y0, y1, y2}, x]
```

Out[37]=
$$y0 + \frac{(x - x0)\ (-y0 + y1)}{-x0 + x1} + \frac{(x - x0)\ (x - x1)\ \left(-\frac{-y0+y1}{-x0+x1} + \frac{-y1+y2}{-x1+x2}\right)}{-x0 + x2}$$

Konkrét Lagrange polinom előállításhoz:

In[38]:=
```
NewtInterp[{1, 2, 3, 4, 5}, {0, 6, 24, 60, 120}, x]
```

Out[38]= $6\ (-1 + x) + 6\ (-2 + x)\ (-1 + x) + (-3 + x)\ (-2 + x)\ (-1 + x)$

In[39]:=
```
Expand[%]
```

Out[39]= $-x + x^3$

Megjegyzés. Séma az f[x0], f[x0;x1], f[x0;x1;x2], f[x0;x1;x2;x3],f[x0;x1;x2;x3;x4] osztott differencia sorozat kiszámításához.

$$L_4[x] = ((((0) + 6\ (-1 + x)) + 6\ (-2 + x)\ (-1 + x)) + 1\ (-3 + x)\ (-2 + x)\ (-1 + x)) + 0\ (-1 + x)\ (-2 + x)\ (-3 + x)\ (-4 + x)\ (-5 + x)$$

x_i	y_i	1 .	2.	3.	4.
1	0				
		6			
2	6		6		
		18		1	
3	24		9		0
		36		1	
4	60		12		
		60			
5	120				

Hermite: Black-Box Feladatmegoldás

Mi a következõ Hermite interpolációs probléma megoldása?

x0={1,2,3,4}
y0={1,2,4,3}
yd0={1,0,0,-1}

Ábrázoljuk az "adatokkal" a polinomot!
Oldjuk meg a feladatot a Mathematica függvénnyel!



Built-in Mathematica függvény

```
?InterpolatingPolynomial
```

In[40]:=
```
InterpolatingPolynomial[{{1, {-1, 1}}, {3, {21, 25}}}, x] // Expand
```

Out[40]= $-2\,x + x^3$

In[41]:=
```
x0 = {1, 2, 3, 4};
y0 = {1, 2, 4, 3};
yd0 = {1, 0, 0, -1};
```

In[44]:=
```
hp = Expand[InterpolatingPolynomial[Transpose[{x0, Transpose[{y0, yd0}]}], x]]
```

Out[44]= $\frac{529}{3} - \frac{5915\,x}{9} + \frac{107383\,x^2}{108} - \frac{42533\,x^3}{54} + \frac{38311\,x^4}{108} - \frac{2464\,x^5}{27} + \frac{337\,x^6}{27} - \frac{19\,x^7}{27}$

Grafikon

In[45]:=
```
Plot[hp, {x, 0, 5}, PlotRange → {-1, 5}, PlotStyle → Thickness[.005]]
```

Out[45]=


Grafikon+pontok

In[46]:=
```
L[{xx_, yy_, ydd_}, var_] := ydd (var - xx) + yy
```

In[47]:=
```
L[{1, 4, 2}, x] // Expand
```

Out[47]= $2 + 2\,x$

In[48]:=
```
ls = Map[L[#, x] &, Transpose[{x0, y0, yd0}]]
```

Out[48]= $\{x, 2, 4, 7 - x\}$

grafikon+pontok+érintőegyenesek

In[49]:=
```
Plot[Evaluate[Prepend[ls, hp]], {x, 0, 5}, PlotRange → {-1, 5},
 PlotStyle → Prepend[Table[{Blue}, {4}], {Thickness[.005], Black}],
 Epilog → {Red, PointSize[.02], Map[Point[#] &, Transpose[{x0, y0}]]}]
```

Out[49]=


grafikon+pontok+érintőegyenes-szakaszok

In[50]:=
```
Plot[hp, {x, 0, 5}, PlotStyle → Black,
 Epilog → {RGBColor[1, 0, 0], PointSize[.02], Map[Point[#] &, Transpose[{x0, y0}]],
   RGBColor[0, 0, 1], Map[Line[{{#[[1]] - .5, #[[2]] - #[[3]] .5}, {#[[1]] + .5, #[[2]] + #[[3]] .5}}] &,
    Transpose[{x0, y0, yd0}]]}]
```

Out[50]=


HDEMO

n

DF: 0 1 F: 2 0		⇑ ⇓ ⇐ ⇒ + I
x	1 3	$2-\frac{1}{2}(-1+x)^2+\frac{3}{4}(-3+x)(-1+x)^2$

Hermite Interpolation 2012

- **Hermite Demo aux**
- **Példa**

A Hermite interpolációs polinomot ismét szukcesszív lineáris egyenletmegoldással keressük meg, lásd a Lagrange-problémánál lévõ algoritmust!

1. Példa p(1)=2, p'(1)=0, p(3)=0, p'(3)=1

azaz X={1,3}
Y={2,0}
YD={0,1}

```
InterpolatingPolynomial[{{1, {2, 0}}, {3, {0, 1}}}, x]
```

$$2+\left(-\frac{1}{2}+\frac{3}{4}(-3+x)\right)(-1+x)^2$$

```
InterpolatingPolynomial[{{1, {2, 0}}, {3, {0, 1}}}, x] // Expand
```

$$-\frac{3}{4}+\frac{25\,x}{4}-\frac{17\,x^2}{4}+\frac{3\,x^3}{4}$$

Poliomfejlesztés és a
A ND-s háromszögtáblázat értelemszerű módosítása

```
Clear[p0, p1, p2, p3]
```

```
p0 = 2;
```

```
p1 = p0 + A (x - 1)
```

$2+A(-1+x)$

```
Solve[(D[p1, x] /. x → 1) == 0, A]
```

{{A → 0}}

```
p1 = 2 + 0 (x - 1) // Expand
```

2

```
p2 = p1 + A (x - 1) (x - 1)
```

$2 + A\,(-1 + x)^2$

```
Solve[(p2 /. x → 3) == 0, A]
```

$\left\{\left\{A \to -\frac{1}{2}\right\}\right\}$

```
p2 = p1 + (-1 / 2) (x - 1)^2
```

$2 - \frac{1}{2}\,(-1 + x)^2$

```
p3 = p2 + A (x - 1)^2 (x - 3)
```

$2 - \frac{1}{2}\,(-1 + x)^2 + A\,(-3 + x)\,(-1 + x)^2$

```
Solve[(D[p3, x] /. x → 3) == 1, A]
```

$\left\{\left\{A \to \frac{3}{4}\right\}\right\}$

```
p3 = p2 + 3/ 4 (x - 1)^2 (x - 3)
```

$2 - \frac{1}{2}\,(-1 + x)^2 + \frac{3}{4}\,(-3 + x)\,(-1 + x)^2$

x_i	f_i	1 .	2.	3.
1	2			
		0		
1	2		-1/ 2	
		-1		3/ 4
3	0		1	
		1		
3	0			

Hermite alappolinomok (a Lagrange alappolinomok felhasználásával)

Az interpolációs probléma megoldása egy legfeljebb (2n-1)-edfokú polinom. Pl. n=3, deg P≤5
Az elôállítás alapgondolata: Kétféle bázispolinom: h_j ill. h_{n+j} alakjuk: $\left(a_j\, x + b_j\right)\, l_j[x]^2$ ill. $\left(x - x_j\right)\, l_j[x]^2$

Mindkettô támaszkodik a Lagrange-féle l_j bázispolinomokra. A feltételek $h_j[x_k] = \delta_{j\,k}$, $h_j{}'[x_k] = 0$, $h_{n+j}[x_k] = 0$, $h_{n+j}{}'[x_k] = \delta_{j\,k}$

6 feltétel automatikusan teljesül, 2 pedig meghatározza a lineáris faktor együtthatóit $\left(1 - 2\left(x - x_j\right)\, l_j{}'[x]\right)$

In[51]:=
$$\text{LagrBase}[j\_, x\_List, var\_] := \left(\prod_{k=1}^{j} \frac{var - x[[k]]}{x[[j+1]] - x[[k]]}\right) \left(\prod_{k=j+2}^{Length[x]} \frac{var - x[[k]]}{x[[j+1]] - x[[k]]}\right)$$

In[52]:=
```
Clear[HermBase]
```

In[53]:=
```
HermBase[j_, x_List, var_] /; j < Length[x] :=
  (1 - 2 (var - x[[j + 1]]) ((D[LagrBase[j, x, var], var]) /. var → x[[j + 1]]))
   LagrBase[j, x, var]^2;

HermBase[j_, x_List, var_] := (var - x[[j - Length[x] + 1]]) LagrBase[j - Length[x], x, var]^2;
```

In[55]:=
$$\text{HermInterp}[x\_List, y\_List, yd\_List, var\_] := \sum_{j=0}^{Length[x]-1} y[[j+1]]\ \text{HermBase}[j, x, var] + \sum_{j=0}^{Length[x]-1} yd[[j+1]]\ \text{HermBase}[Length[x]+j, x, var]$$

■ Feladat

x={x0,x1,x2}={1,2,3}
y={y0,y1,y2}={2,32,242}
yd={yd0,yd1,yd2}={4,79,404}

Adjuk meg a Hermite alappolinomokat és a Hermite interpolációs polinomot!
Ellenõrizzük a nevezetes tulajdonságokat!

■ Javaslat

In[56]:=
```
x0 = {1, 2, 3};
y0 = {2, 32, 242};
yd = {4, 79, 404};
```

In[59]:= `HermBase[3, x0, x]`

Out[59]= $\frac{1}{4} (2 - x)^2 (3 - x)^2 (-1 + x)$

In[60]:= `LagrBase[0, x0, x]`

Out[60]= $\frac{1}{2} (2 - x) (3 - x)$

In[61]:= `HermBase[0, x0, x]`

Out[61]= $\frac{1}{4} (1 + 3 (-1 + x)) (2 - x)^2 (3 - x)^2$

In[62]:= `q = (a x + b) LagrBase[0, x0, x]^2`

Out[62]= $\frac{1}{4} (2 - x)^2 (3 - x)^2 (b + a x)$

In[63]:= `q /. x -> x0[[1]]`

Out[63]= a + b

In[64]:= `D[q, x] /. x -> x0[[1]]`

Out[64]= $-b - 2 (a + b)$

```
In[65]:= q
```

Out[65]= $\frac{1}{4}(2-x)^2(3-x)^2(b+a\,x)$

```
In[66]:= Solve[{(q /. x -> x0[[1]]) == 1, (D[q, x] /. x -> x0[[1]]) == 0}, {a, b}]
```

Out[66]= $\{\{a\to 3,\ b\to -2\}\}$

```
In[67]:= q /. Solve[{(q /. x -> x0[[1]]) == 1, (D[q, x] /. x -> x0[[1]]) == 0}, {a, b}][[1]] // Expand
```

Out[67]= $-18+57\,x-\frac{127\,x^2}{2}+\frac{131\,x^3}{4}-8\,x^4+\frac{3\,x^5}{4}$

```
In[68]:= HermBase[0, x0, x] // Expand
```

Out[68]= $-18+57\,x-\frac{127\,x^2}{2}+\frac{131\,x^3}{4}-8\,x^4+\frac{3\,x^5}{4}$

```
In[69]:=
H = Table[HermBase[j, x0, x], {j, 0, 2 Length[x0] - 1}]
```

Out[69]= $\left\{\frac{1}{4}(1+3(-1+x))(2-x)^2(3-x)^2,\ (3-x)^2(-1+x)^2,\ \frac{1}{4}(1-3(-3+x))(-2+x)^2(-1+x)^2,\ \frac{1}{4}(2-x)^2(3-x)^2(-1+x),\ (3-x)^2(-2+x)(-1+x)^2,\ \frac{1}{4}(-3+x)(-2+x)^2(-1+x)^2\right\}$

```
In[70]:= Length[%]
```

Out[70]= 6

A nulladik és a harmadik bázispolinomra kirótt feltételek vizsgálata

```
In[71]:= h0[x_] = Expand[HermBase[0, x0, x]];
In[72]:= h3[x_] = Expand[HermBase[3, x0, x]];
In[73]:= h3[x0[[3]]]
Out[73]= 0
In[74]:= hd0[x_] = D[HermBase[0, x0, x], x];
In[75]:= hd3[x_] = D[HermBase[3, x0, x], x];
In[76]:= {{h0[1], h0[2], h0[3]}, {hd0[1], hd0[2], hd0[3]}}
Out[76]= {{1, 0, 0}, {0, 0, 0}}
In[77]:= {{h3[1], h3[2], h3[3]}, {hd3[1], hd3[2], hd3[3]}}
Out[77]= {{0, 0, 0}, {1, 0, 0}}
```

Az elsô és a negyedik bázispolinomra kirótt feltételek vizsgálata

```
In[78]:= h1[x_] = Expand[HermBase[1, x0, x]];
In[79]:= h4[x_] = Expand[HermBase[4, x0, x]];
In[80]:= hd1[x_] = D[HermBase[1, x0, x], x];
In[81]:= hd4[x_] = D[HermBase[4, x0, x], x];
In[82]:= {{h3[1], h3[2], h3[3]}, {hd3[1], hd3[2], hd3[3]}}
Out[82]= {{0, 0, 0}, {1, 0, 0}}
```

```
In[83]:= {{h1[1], h1[2], h1[3]}, {hd1[1], hd1[2], hd1[3]}}
Out[83]= {{0, 1, 0}, {0, 0, 0}}
In[84]:= {{h4[1], h4[2], h4[3]}, {hd4[1], hd4[2], hd4[3]}}
Out[84]= {{0, 0, 0}, {0, 1, 0}}
In[85]:= h2[x_] = Expand[HermBase[2, x0, x]];
In[86]:= h5[x_] = Expand[HermBase[5, x0, x]];
In[87]:= hd2[x_] = D[HermBase[2, x0, x], x];
In[88]:= hd5[x_] = D[HermBase[5, x0, x], x];
In[89]:= {{h2[1], h2[2], h2[3]}, {hd2[1], hd2[2], hd2[3]}}
Out[89]= {{0, 0, 1}, {0, 0, 0}}
```

Hermite Interpolációs Polinom

```
In[90]:= HermInterp[x0, y0, yd, x]
```

Out[90]=
$$\frac{1}{2}(1+3(-1+x))(2-x)^2(3-x)^2+(2-x)^2(3-x)^2(-1+x)+32(3-x)^2(-1+x)^2+79(3-x)^2(-2+x)(-1+x)^2+\frac{121}{2}(1-3(-3+x))(-2+x)^2(-1+x)^2+101(-3+x)(-2+x)^2(-1+x)^2$$

Fokszám: 5

```
In[91]:= Expand[HermInterp[x0, y0, yd, x]]
```

Out[91]= $2-x+x^5$

In[92]:= p1 = 2 - x + x^5;

```
In[93]:= D[p1, x] /. x → 3
Out[93]= 404
In[94]:= x0
Out[94]= {1, 2, 3}
In[95]:= y0
Out[95]= {2, 32, 242}
In[96]:= Clear[p2]
```

In[97]:= p2[x_] := 2 - x + x^5;

```
In[98]:= p2[3]
Out[98]= 242
```